stata.com

power trend — Power analysis for the Cochran-Armitage trend test

Description Options References Quick start Remarks and examples Also see Menu Stored results Syntax Methods and formulas

## Description

Title

power trend computes sample size or power for the Cochran-Armitage trend test, a test for a linear trend in a probability of response in  $J \times 2$  tables. It can accommodate unbalanced designs and unequally spaced exposure levels (doses). With equally spaced exposure levels, a continuity correction is available.

Sample size for a test with alternative probabilities of 0.2, 0.3, and 0.4 using default power of 0.8

## Quick start

and significance level  $\alpha = 0.05$ power trend .2 .3 .4 As above, but for power of 0.9 power trend .2 .3 .4, power(.9) As above, but for power of 0.7, 0.75, 0.8, 0.85, and 0.9 power trend .2 .3 .4, power(.7(.05).9) Sample size for a one-sided test power trend .2 .3 .4, onesided As above, and apply continuity correction power trend .2 .3 .4, onesided continuity Power for a total sample size of 240 subjects power trend .2 .3 .4, n(240) Same as above, specified as 3 groups of 80 subjects each power trend .2 .3 .4, npergroup(80) Power for 100 subjects in group 1, 80 in group 2, and 60 in group 3 power trend .2 .3 .4, n1(100) n2(80) n3(60) Graph of power against group sample size for group sizes of 70, 80, 90, and 100

power trend .2 .3 .4, npergroup(70(10)100) graph

### Menu

Statistics > Power and sample size

## Syntax

Compute sample size power trend probspec [, power(numlist) options]

Compute power

power trend probspec, n(numlist) [ options ]

where *probspec* is either a matrix *matname* containing group probabilities or a list of individual group probabilities:

 $p_1 p_2 \left[ p_3 \ldots p_J \right]$ 

 $p_j$ , where j = 1, 2, ..., J, is the alternative group probability of observing a success for subjects with the *j*th level of exposure. Each  $p_j$  may be specified either as one number or as a list of values in parentheses; see [U] **11.1.8 numlist**.

*matname* is the name of a Stata matrix with J columns containing values of alternative group probabilities. Multiple rows are allowed, in which case each row corresponds to a different set of J group probabilities or, equivalently, column j corresponds to a *numlist* for the jth group probabilities.

Alternative probabilities should be strictly monotonic: all increasing or all decreasing.

#### power trend — Power analysis for the Cochran-Armitage trend test 3

options	Description
Main	
* <u>a</u> lpha( <i>numlist</i> )	significance level; default is alpha(0.05)
* power(numlist)	power; default is power(0.8)
* beta( <i>numlist</i> )	probability of type II error; default is beta(0.2)
* n(numlist)	total sample size; required to compute power
<u>nfrac</u> tional	allow fractional sample sizes
*npergroup(numlist)	number of subjects per group; implies balanced design
* n#(numlist)	number of subjects in group #
<pre>grweights(wgtspec)</pre>	group weights; default is one for each group, meaning equal group sizes
exposure(exposspec)	strictly increasing exposure levels; default is equally spaced ordinal values
<u>contin</u> uity	apply the continuity correction; default is no continuity correction
<u>onesid</u> ed	one-sided test; default is two sided
<u>par</u> allel	treat number lists in starred options or in command arguments as parallel when multiple values per option or argument are specified (do not enumerate all possible combinations of values)
Table	
$[\underline{no}]\underline{tab}le[(tablespec)]$	suppress table or display results as a table; see [PSS] <b>power, table</b>
<pre>saving(filename [, replace])</pre>	save the table data to <i>filename</i> ; use replace to overwrite existing <i>filename</i>
Graph	
$\underline{graph}[(graphopts)]$	graph results; see [PSS] power, graph
Iteration	
<pre>init(#)</pre>	initial value for the sample size for a two-sided test; default is to use a sample-size estimate for a one-sided test
<u>iter</u> ate(#)	maximum number of iterations; default is iterate(500)
<u>tol</u> erance(#)	parameter tolerance; default is tolerance(1e-12)
<pre>ftolerance(#)</pre>	function tolerance; default is ftolerance(1e-12)
[no]log	suppress or display iteration log
[no]dots	suppress or display iterations as dots
notitle	suppress the title

\*Specifying a list of values in at least two starred options, or at least two command arguments, or at least one starred option and one argument results in computations for all possible combinations of the values; see [U] 11.1.8 numlist. Also see the parallel option.

notitle does not appear in the dialog box.

wgtspec	Description
$\overline{\#_1 \ \#_2 \ \ldots \ \#_J}$	J group weights. Weights must be positive and must be integers unless option nfractional is specified. Multiple values for each group weight $\#_j$ can be specified as a <i>numlist</i> enclosed in parentheses.
matname	matrix with $J$ columns containing $J$ group weights. Multiple rows are allowed, in which case each row corresponds to a different set of $J$ weights or, equivalently, column $j$ corresponds to a <i>numlist</i> for the $j$ th weight.
exposspec	Description
$\overline{\#_1 \ \#_2 \ \ldots \ \#_J}$	J exposure levels. By default, equally spaced exposure levels of 1, 2,, J are used. Multiple values for each exposure level $\#_j$ can be specified as a <i>numlist</i> enclosed in parentheses.
matname	matrix with $J$ columns containing $J$ exposure levels. Multiple rows are allowed; in which case each row corresponds to a different set of $J$ exposure levels or, equivalently, column $j$ corresponds to a <i>numlist</i> for the $j$ th exposure level.

where tablespec is

column[:label] [column[:label] [...]] [, tableopts]

*column* is one of the columns defined below, and *label* is a column label (may contain quotes and compound quotes).

column	Description	Symbol
alpha	significance level	α
power	power	$1 - \beta$
beta	type II error probability	$\beta$
Ν	total number of subjects	N
N_per_group	number of subjects per group	$N/N_g$
N_avg	average number of subjects per group	$N_{\mathrm{avg}}$
N#	number of subjects in group #	$N_{\#}$
N_g	number of groups	$N_{g}$
p#	probability of outcome for group #	$p_{\#}$
x#	exposure level #	x <sub>#</sub>
grwgt#	group weight #	$w_{\#}$
_all	display all supported columns	

Column beta is shown in the default table in place of column power if option beta() is specified.

Column N\_per\_group is shown in the default table only for balanced designs.

Columns N\_avg and N# are shown in the default table only for unbalanced designs.

Columns x# are shown only when exposure levels are specified using the exposure() option.

# Options

\_ Main \_

alpha(), power(), beta(), n(), nfractional; see [PSS] power.

- n#(numlist) specifies the number of subjects in the #th group to be used for power determination. Only positive integers are allowed. All group sizes must be specified. n#() cannot be specified with n(), npergroup(), or grweights().
- npergroup(numlist) specifies the group size. Only positive integers are allowed. This option implies
   a balanced design. npergroup() cannot be specified with n(), n#(), or grweights().
- grweights (*wgtspec*) specifies J group weights for an unbalanced design. The weights may be specified either as a list of values or as a matrix, and multiple sets of weights are allowed; see *wgtspec* for details. The weights must be positive and must also be integers unless the nfractional option is specified. grweights() cannot be specified with npergroup() or n#().
- exposure (*exposspec*) specifies the J strictly increasing exposure levels. The default is to use equally spaced values of 1, 2, ..., J.
- continuity requests that the continuity correction be applied. This option can be specified only for equally spaced exposure levels. By default, no continuity correction is applied.

onesided, parallel; see [PSS] power.

Table ]

table, table(), notable; see [PSS] power, table.

saving(); see [PSS] power.

Graph

graph, graph(); see [PSS] power, graph. Also see the *column* table for a list of symbols used by the graphs.

\_\_\_\_ Iteration

init(#) specifies the initial value of the sample size for the sample-size computation for a two-sided test. The default initial value is the sample size for the corresponding one-sided test.

iterate(), tolerance(), ftolerance(), log, nolog, dots, nodots; see [PSS] power.

The following option is available with power trend but is not shown in the dialog box:

notitle; see [PSS] power.

## **Remarks and examples**

stata.com

Remarks are presented under the following headings:

Introduction Using power trend Alternative ways of specifying probabilities Computing sample size Computing power Testing hypotheses about a trend in J×2 tables This entry describes the power trend command and the methodology for power and sample-size analysis for Cochran-Armitage test for a linear trend in probability of response in  $J \times 2$  tables. See [PSS] intro for a general introduction to power and sample-size analysis and [PSS] power for a general introduction to the power command using hypothesis tests.

#### Introduction

Studies that examine the relationship between an exposure and a binary outcome have many biomedical and social science applications. When exposure can be treated as ordinal levels, researchers are often interested in whether there is a trend, or dose-response relationship, in exposure and a binary outcome. The data are typically summarized in an ordered  $J \times 2$  table,

	Binary response			
Exposure level	Success	Failure		
$\begin{array}{c} x_1 \\ x_2 \end{array}$	$\begin{array}{c} m_1 \\ m_2 \end{array}$	$n_1 - m_1 \\ n_2 - m_2$		
:	:	:		
$x_J$	$m_J$	$n_J - m_J$		

where  $x_j$  is an ordinal level (table score) or rank score associated with the exposure (dose) received by group j such that  $x_{j-1} < x_j$ ,  $m_j$  is the number of successes in group j and  $n_j$  is the number of subjects in group j for each j = 1, 2, ..., J. For equally spaced exposure levels, the levels are often assigned ordinal numbers;  $x_j = j$ , j = 1, 2, ..., J.

A "success" simply means observing an event of interest. A dermatologist might wish to identify dosage of a topical antibiotic (ordinal exposure) necessary to cure a skin infection (binary outcome). A oncologist might conduct a case–control study to see if the number of first- and second-degree relatives with a BRCA1 gene mutation is associated with the occurrence of breast cancer. An education researcher might want to know if a higher number of unexcused absences from school is associated with failing a school grade.

This entry describes power and sample-size analysis for inference using hypothesis testing about the presence of a linear trend in probability of response in  $J \times 2$  tables. The Cochran–Armitage trend test (Cochran 1954 and Armitage 1955) is commonly used to test for trend in  $J \times 2$  tables. It is based on the linear logit model,

$$logit(p_i) = a + bx_i$$

where  $p_j$  is the hypothesized probability of a success in group j, and a and b are unknown coefficients. These group probabilities can be estimated from our contingency table as  $\hat{p}_j = m_j/n_j$ .

The null hypothesis of interest is  $H_0: p_1 = p_2 = \cdots = p_J$  against the one-sided increasing-trend alternative  $H_a: p_1 < p_2 < \cdots < p_J$ , the one-sided decreasing-trend alternative  $H_a: p_1 > p_2 > \cdots > p_J$ , or the two-sided alternative  $H_a: p_1 < p_2 < \cdots < p_J$  or  $p_1 > p_2 > \cdots > p_J$ . For the linear logit model, these hypothesis are equivalent to the null  $H_0: b = 0$  against the two-sided alternative  $H_a: b \neq 0$ . If we believe that the probability of a success increases with exposure, the alternative hypothesis is upper one-sided,  $H_a: b > 0$ . If we believe that the probability of a success decreases with exposure, the alternative hypothesis is lower one-sided,  $H_a: b < 0$ . The test statistic for testing  $H_0: b = 0$  is asymptotically normal under the null hypothesis.

Power and sample-size computations are based on the asymptotic distribution of the test statistic.

#### Using power trend

power trend computes sample size or power for a Cochran-Armitage trend test in  $J \times 2$  tables. All computations are performed for a two-sided hypothesis test where, by default, the significance level is set to 0.05. You may change the significance level by specifying the alpha() option. You can specify the onesided option to request a one-sided test.

To compute the total and individual group sample sizes, you must specify the alternative probabilities of a success for J levels of exposure and, optionally, the power of the test in the power() option. The default power is set to 0.8.

To compute power, you must specify the total sample size in the n() option and the alternative probabilities.

There are multiple ways to specify the alternative probabilities; see Alternative ways of specifying probabilities.

By default, all computations assume a balanced- or equal-allocation design. You can use the grweights() option to specify an unbalanced design for power or sample-size computations. For power computations, you can specify individual group sizes in options n1(), n2(), ..., nJ() instead of a combination of n() and grweights() to accommodate an unbalanced design. For a balanced design, you can also specify the npergroup() option to specify a group size instead of a total sample size in n().

Computations also assume that exposure levels are equally spaced and no continuity correction is applied. When the exposure levels are equally spaced, you can use option continuity to request that the continuity correction be applied. You may specify specific exposure levels in the exposure() option. There are multiple ways of specifying the levels; any method described in *Alternative ways of specifying probabilities* can also be applied to the specification of exposure levels.

Sample-size determination for a two-sided test requires iteration. The default initial values are sample-size estimates for the corresponding one-sided test. You can use the init() option to specify your own value. See [PSS] **power** for a description of other options that control the iteration process.

#### Alternative ways of specifying probabilities

There are multiple ways in which you can supply the group probabilities of success to power trend.

You may specify each  $p_i$  following the command line as

power trend  $p_1 p_2 \ldots p_J$  [, ...]

At least two probabilities must be specified.

When you have many groups, you may find it more convenient to first define a Stata matrix as a column vector and use it with power trend. The dimension of the matrix must be at least 2. For example,

```
matrix define probmat = (p_1, p_2, ..., p_J)
power trend probmat [, ...]
```

In some cases, you may wish to examine multiple alternative probabilities for one or more groups. To do this, you can specify multiple values or a *numlist* for each of the group probabilities in parentheses.

power trend  $(p_{1,1} \ p_{1,2} \ \dots \ p_{1,K_1})$   $(p_{2,1} \ p_{2,2} \ \dots \ p_{2,K_2})$   $\dots$   $[, \ldots]$ 

Each of the *numlists* may contain different numbers of values,  $K_1 \neq K_2 \neq \cdots \neq K_J$ . power trend will produce results for all possible combinations of values across *numlists*. Results are presented in a table. If instead you would like to treat each specification as a separate scenario, you may specify the parallel option.

You can accommodate multiple sets of group probabilities in a matrix form by adding a row for each specification. The columns of a matrix with multiple rows correspond to J group probabilities, and values within each column j correspond to multiple specifications of the jth group probability of a success or a *numlist* for the jth group probability.

For example, the following two specifications for three groups defined by their exposure levels with two scenarios each are the same:

```
power trend (p_{1,1} \ p_{1,2}) (p_{2,1} \ p_{2,2}) (p_{3,1} \ p_{3,2}) [, ...]
```

and

```
matrix define probmat = (p_{1,1}, p_{2,1}, p_{3,1} \setminus p_{1,2}, p_{2,2}, p_{3,2})
```

```
power trend probmat [, ...]
```

In the above specification, if you wish to specify a *numlist* only for the first group, you may define your matrix as

matrix define probmat =  $(p_{1,1}, p_{2,1}, p_{3,1} \setminus p_{1,2}, ..., .)$ 

and the results of

```
power trend probmat [, ...]
```

will be the same as the results of

power trend (  $p_{1,1} \ p_{1,2}$  )  $p_{2,1} \ p_{3,1}$  [ ,  $\ldots ]$ 

In the following sections, we describe the use of power trend accompanied by examples for computing sample size and power.

#### Computing sample size

To compute sample size, you must specify the alternative probabilities of a success for each of the J exposure levels and, optionally, the power of the test in the power() option. A default power of 0.8 is assumed if power() is not specified.

#### Example 1: Sample size for a two-sided trend test

Consider a study investigating the effectiveness of a new topical antibiotic for the treatment of skin infections.

Suppose that in previous studies of the treatment, we observed the following proportions of successfully treated cases at different doses. We may hypothesize that these represent the probability of a successful treatment for each dose.

Doses/day	Proportion Successes
1	0.80
2	0.85
3	0.90

We wish to determine the minimum sample size required for a clinical trial designed to detect a dose–response trend with 80% power using a two-sided 5%-level test.

To compute the required sample size, we specify the values 0.80, 0.85, and 0.90 as the alternative probabilities after the command name. We omit options alpha(0.05) and power(0.8) because the specified values are their defaults.

```
. power trend .80 .85 .90
note: exposure levels are assumed to be equally spaced
Performing iteration ...
Estimated sample size for a trend test
Cochran-Armitage trend test
Ho: b = 0 versus Ha: b != 0; logit(p) = a + b*x
Study parameters:
        alpha =
                   0.0500
        power =
                   0.8000
                        3
          N_g =
           p1 =
                   0.8000
           p2 =
                   0.8500
           p3 =
                   0.9000
Estimated sample sizes:
            N =
                      597
                      199
  N per group =
```

A total sample of 597 individuals, 199 individuals per group, must be obtained to detect a linear trend in probability of a successful treatment with 80% power using a two-sided 5%-level Cochran–Armitage test.

4

#### Example 2: Sample size for a one-sided test

Continuing with example 1, suppose that the relevant research question is whether the probability of a successful treatment increases with the number of doses. In this case, we would choose to use a one-sided test because we are only interested in an increasing trend.

By specifying the onesided option, we obtain the sample size needed to detect a positive trend with 80% power using a one-sided 5%-level Cochran–Armitage test.

. power trend 0.80 0.85 0.90, onesided note: exposure levels are assumed to be equally spaced Estimated sample size for a trend test Cochran-Armitage trend test Ho: b = 0 versus Ha: b > 0; logit(p) = a + b\*x Study parameters: alpha = 0.0500 power = 0.8000  $N_g =$ 3 p1 = 0.8000 p2 = 0.8500 p3 = 0.9000 Estimated sample sizes: N = 471 N per group = 157

Switching to a one-sided hypothesis decreased our total sample-size requirement to 471 individuals. Because the doses are equally spaced, it is possible to also perform a continuity correction. If we wanted this correction, we would have also specified the continuity option.

4

#### Example 3: Unbalanced design

Continuing with example 1, we have assumed that the participants will be equally divided among the treatment groups. Thus, we would randomize 199 participants to each treatment level. Suppose that we instead plan to have twice as many subjects at the lowest treatment level. We can accommodate this unbalanced design by specifying the corresponding group weights in the grweights() option.

```
. power trend 0.80 0.85 0.90, grweights(2 1 1)
note: exposure levels are assumed to be equally spaced
Performing iteration ...
Estimated sample size for a trend test
Cochran-Armitage trend test
Ho: b = 0 versus Ha: b != 0; logit(p) = a + b*x
Study parameters:
        alpha =
                   0.0500
        power =
                   0.8000
                        3
          N_g =
           p1 =
                   0.8000
           p2 =
                   0.8500
                   0.9000
           p3 =
Estimated sample sizes:
            N =
                      600
    Average N =
                 200.0000
           N1 =
                      300
           N2 =
                      150
           N3 =
                      150
```

The required total sample size for this unbalanced design is 600 with 300 subjects in the group receiving a single dose per day and 150 in the groups receiving two and three doses per day. The average number of subjects per group is 200.

#### Example 4: Sample size for unequally spaced exposures

It is possible that the groups do not represent equally spaced exposures. Consider an example from Agresti (2013, 89) on the association between maternal drinking and congenital malformations, which shows data originally from Graubard and Korn (1987). The data are reported in ranges rather than actual values.

```
. use http://www.stata-press.com/data/r14/infants
(Congenital Malformation Data)
. list, noobs abbreviate(12)
```

consump	cscore	infants	cases	prmalform
0	0.0	17,114	48	0.0028
< 1	0.5	14,502	38	0.0026
1 to 2	1.5	793	5	0.0063
3 to 5	4.0	127	1	0.0079
>= 6	7.0	38	1	0.0263

Suppose we wish to use these data to design a new study to test whether there is an increasing trend in the probability of congenital malformation as the average number of alcoholic beverages consumed each week by the mother increases. We will use the onesided option as we did in example 2 to obtain sample size with 80% power using a one-sided 5%-level Cochran–Armitage test.

To accurately compute the sample size that we will need, we should use the score associated with the reported range. Agresti recommends using the midpoint of the range for a Cochran–Armitage trend test and adopts an arbitrary value of 7 for the last interval in this case.

We can relax the assumption that the exposure levels for alcohol consumption are evenly spaced by adding the exposure() option. Because we have 5 exposure levels, we use a matrix specification of alternative probabilities and exposure levels. We can use the mkmat command to create the matrices rather than retyping the values that we already have in our dataset; see [P] matrix mkmat.

```
. mkmat prmalform, matrix(p)
. matrix list p
p[5,1]
    prmalform
r1
    .00280472
r2
    .00262033
r3
    .00630517
r4
    .00787402
   .02631579
r5
. mkmat cscore, matrix(exposed)
. matrix list exposed
exposed[5,1]
    cscore
r1
         0
r2
        .5
r3
       1.5
r4
         4
         7
r5
```

Now, we specify the matrix of probabilities p after the command power trend and the matrix of exposure levels exposed in option exposure().

```
. power trend p, onesided exposure(exposed)
Estimated sample size for a trend test
Cochran-Armitage trend test
Ho: b = 0 versus Ha: b > 0; logit(p) = a + b*x
Study parameters:
        alpha =
                   0.0500
        power =
                   0.8000
          N_g =
                        5
                   0.0028
                                     x1 =
           p1 =
                                             0.0000
           p2 =
                   0.0026
                                     x2 =
                                             0.5000
           p3 =
                   0.0063
                                     x3 =
                                             1.5000
           p4 =
                   0.0079
                                     x4 =
                                             4.0000
           p5 =
                   0.0263
                                     x5 =
                                             7.0000
Estimated sample sizes:
            N =
                      1030
  N per group =
                      206
Warning: Alternative probabilities are not monotonic.
```

To conduct this study, we would need to recruit a total sample of 1,030 mothers with 206 mothers at each level of consumption.

For this example, power trend reported a warning message that the specified alternative probabilities are not monotonic. The power for the test depends on the specific alternative, which is  $p_1 < p_2 < \cdots < p_J$  in this example. Gross departures from the monotonicity assumption may lead to invalid results. In our example, the violation of this assumption is very mild—the offending probabilities are  $p_1 = 0.0028$  and  $p_2 = 0.0026$ .

4

#### Computing power

To compute power, you must specify the alternative probabilities after the command name and the total sample size in the n() option.

#### Example 5: Power of a two-sided Cochran–Armitage trend test

Returning to example 1, suppose that we anticipate obtaining a sample size of only 540 participants. To compute the corresponding power, we specify the sample size of 540 in n():

```
. power trend 0.80 0.85 0.90, n(540)
note: exposure levels are assumed to be equally spaced
Estimated power for a trend test
Cochran-Armitage trend test
Ho: b = 0 versus Ha: b != 0; logit(p) = a + b*x
Study parameters:
        alpha =
                   0.0500
            N =
                      540
  N per group =
                      180
          N_g =
                         3
           p1 =
                   0.8000
           p2 =
                   0.8500
           рЗ =
                   0.9000
Estimated power:
        power =
                   0.7592
```

Power decreases to 75.9% with the smaller sample of 540 subjects.

#### Example 6: Multiple values of study parameters

We may want to check powers for several sample sizes. Continuing with example 5, we simply list sample-size values in the option n(numlist); see [U] 11.1.8 numlist.

. power trend 0.80 0.85 0.90, n(540 570 600 630 660)
> table(, labels(N\_per\_group "N/N\_g") formats("%6.2g"))
note: exposure levels are assumed to be equally spaced
Estimated power for a trend test
Cochran-Armitage trend test
Ho: b = 0 versus Ha: b != 0; logit(p) = a + b\*x

alpha	power	N	N/N_g	N_g	p1	p2	рЗ
.05	.76	540	180	3	.8	.85	.9
.05	.78	570	190	3	.8	.85	.9
.05	.8	600	200	3	.8	.85	.9
.05	.82	630	210	3	.8	.85	.9
.05	.84	660	220	3	.8	.85	.9

To shorten our default table, we specified a shorter label for the N\_per\_group column and reduced the default display format for all table columns by specifying the corresponding options within the table() option.

For multiple values of parameters, the results are automatically displayed in a table, as we see above. For more examples of tables, see [PSS] **power, table**. If you wish to produce a power plot, see [PSS] **power, graph**.

4

#### Testing hypotheses about a trend in $J \times 2$ tables

There are several ways to conduct a trend test in Stata. We demonstrate one method here. For more examples showing the Cochran–Armitage trend test in Stata, see Sribney (1996).

#### Example 7: Testing hypotheses about trends

Returning to example 4, let's test whether the proportion of infants with congenital malformations increases as maternal alcohol consumption per week increases using Agresti's data. (We converted these data from wide to long as needed by further analysis.)

. use http://www.stata-press.com/data/r14/infants2 (Congenital Malformation Data, long form)

. list, noobs sepby(consump)

consump	cscore	malform n
0	0.0	Has malformation 17066
0	0.0	No malformation 48
< 1	0.5	Has malformation 14464
< 1	0.5	No malformation 38
1 to 2 1 to 2	1.5 1.5	
3 to 5	4.0	Has malformation 126
3 to 5	4.0	No malformation 1
>= 6	7.0	Has malformation 37
>= 6	7.0	No malformation 1

In Stata, we can conduct a trend test by using the tabodds command; see [R] epitab.

. tabodds malform cscore [fweight=n]

cscore	cases	controls	0	dds	[95% Conf.	Interval]
0 .5 1.5 4 7	48 38 5 1 1	17066 14464 788 126 37	0.00 0.00 0.00 0.00 0.02	263 635 794	0.00212 0.00191 0.00263 0.00111 0.00371	0.00373 0.00361 0.01529 0.05678 0.19698
·	geneity (equal	Pr>c	hi2 = (1) =	12.08 0.0168 6.57 0.0104		

tabodds reports a  $\chi^2$  value of 6.57 for the trend test. We reject the null hypothesis of no trend at the 5% significance level; the *p*-value is 0.0104.

Note that the  $\chi^2$  statistic reported by tabodds is for a score test for trend of odds. This test is asymptotically equivalent to the Cochran–Armitage trend test. See Lachin (2011) and Agresti (2013) for details.

## Stored results

power trend stores the following in r():

Scalars

Scalars	
r(alpha)	significance level
r(power)	power
r(beta)	probability of a type II error
r(delta)	effect size
r(N)	total sample size
r(N_a)	actual sample size
r(N_avg)	average sample size
r(N#)	number of subjects in group #
r(N_per_group)	number of subjects per group
r(N_g)	number of groups
r(nfractional)	1 if nfractional is specified; 0 otherwise
r(balanced)	1 for a balanced design; 0 otherwise
r(grwgt#)	group weight #
r(onesided)	1 for a one-sided test; 0 otherwise
r(p#)	probability of a success in group #
r(x#)	exposure level for group #
r(continuity)	1 if continuity correction is used; 0 otherwise
r(c)	continuity-correction value
r(separator)	number of lines between separator lines in the table
r(divider)	1 if divider is requested in the table; 0 otherwise
r(init)	initial value for the sample size for a two-sided test
r(maxiter)	maximum number of iterations
r(iter)	number of iterations performed
r(tolerance)	requested parameter tolerance
r(deltax)	final parameter tolerance achieved
r(ftolerance)	requested distance of the objective function from zero
r(function)	final distance of the objective function from zero
r(converged)	1 if iteration algorithm converged; 0 otherwise
Macros	
r(type)	test
r(method)	trend
r(columns)	displayed table columns
r(labels)	table column labels
r(widths)	table column widths
r(formats)	table column formats
Matrix	
r(pss_table)	table of results

# Methods and formulas

Assume that the probability of a success or a positive response,  $p_j$ , follows a linear trend on the logistic scale

$$p_j = \frac{e^{a+bx_j}}{1+e^{a+bx_j}}$$

such that  $logit(p_j) = a + bx_j$ , where  $x_j$  denotes the exposure level (dose) for each of the J groups and  $x_{j-1} < x_j$  for j = 1, 2, ..., J. b is the trend parameter about which we form our hypotheses. Under the null hypothesis,  $H_0$ : b = 0.

Power and sample-size computations for this test are based on Nam (1987).

Methods and formulas are presented under the following headings:

Computing power Computing sample size

#### Computing power

Let  $n_j$  be the sample size at each exposure level for j = 1, 2, ..., J and  $n = \sum_{j=1}^{J} n_j$  denote the total sample size. The observed number of successes  $m_j$  at each of the J exposure levels follows a binomial distribution; the  $m_j$ s are assumed to be independent. Let  $M = \sum_{j=1}^{J} m_j$  denote the total response. Then, the average response rate is given by  $\overline{p} = M/n$  and the average nonresponse rate by  $\overline{q} = 1 - \overline{p}$ .

Let  $U = \sum_{j=1}^{J} m_j x_j$  denote the total exposure-weighted response and  $\overline{x} = \sum_{j=1}^{J} n_j x_j / n$  be the average exposure level in the sample. Denote the conditional mean of U given M as  $E_0(U|M)$ , where  $E_0(U|M) = \overline{p}(\sum_{j=1}^{J} n_j x_j)$ . Let  $U' = U - E_0(U|M) = \sum_{j=1}^{J} m_j (x_j - \overline{x})$ .

Given the significance level  $\alpha$  and the probability of a type II error  $\beta$ , the power  $\pi = 1 - \beta$  is computed as

$$\pi = \begin{cases} \Phi(u_l) & \text{for a lower one-sided test} \\ 1 - \Phi(u_u) & \text{for an upper one-sided test} \\ 1 - \Phi(u_u) + \Phi(u_l) & \text{for a two-sided test} \end{cases}$$
(1)

where  $\Phi(\cdot)$  is the cumulative distribution function of the standard normal distribution. For a one-sided test,

$$u_l = \frac{-E(U'+c) + z_\alpha \sqrt{\operatorname{Var}_0(U')}}{\sqrt{\operatorname{Var}(U')}}$$
(2a)

and

$$u_u = \frac{-E(U'-c) - z_\alpha \sqrt{\operatorname{Var}_0(U')}}{\sqrt{\operatorname{Var}(U')}}$$
(2b)

In (2a) and (2b),  $z_{\alpha}$  is the ( $\alpha$ )th quantile of the standard normal distribution and  $c = (x_{j+1} - x_j)/2$ in the presence of continuity correction or c = 0 in the absence of continuity correction. The correction is not available with unequally spaced exposure levels. The power of the two-sided test uses  $\alpha/2$  to determine the value of z when computing  $u_l$  in (2a) and  $u_u$  in (2b).

Under the null hypothesis,  $p_1 = p_2 = \cdots = p_J$ . Therefore, we can define a common  $p = \sum_{j=1}^J n_j p_j / N$  and q = 1 - p. The variance of U' under the null is

$$\operatorname{Var}_{0}(U') = \operatorname{Var}(U'|H_{0}) = pq \sum_{j=1}^{J} n_{j} (x_{j} - \overline{x})^{2}$$

and the variance of U' given  $p_1, p_2, \ldots, p_J$  is

$$\operatorname{Var}(U') = \sum_{j=1}^{J} n_j p_j q_j (x_j - \overline{x})^2$$

#### Computing sample size

Let  $n_1$  denote the sample size of the reference group, the group with the lowest exposure level, and  $r_j = n_j/n_1$  be the known ratio of sample size of the *j*th group to that of the reference group. Define  $A = \sum_{j=1}^{J} r_j p_j (x_j - \overline{x})$ , and then sample size for a one-sided test without continuity correction is given by

$$n_{1} = \frac{1}{A^{2}} \left[ -z_{\alpha} \sqrt{pq \left\{ \sum_{j=1}^{J} r_{j}(x_{j} - \overline{x})^{2} \right\}} + z_{1-\beta} \sqrt{\sum_{j=1}^{J} r_{j} p_{j} q_{j}(x_{j} - \overline{x})^{2}} \right]^{2}$$
(3)

where  $z_{\alpha}$  is the ( $\alpha$ )th quantile of the standard normal distribution.

The total sample size is computed as  $n = \sum_{j=1}^{J} n_j = n_1 \sum_{j=1}^{J} r_j$ .

The sample-size estimate  $n_{1c}$  for the continuity-corrected statistic is found according to Nam (1987), as follows:

$$n_{1c} = \begin{cases} \frac{n_1}{4} \left( 1 + \sqrt{1 - \frac{4c}{An_1}} \right)^2 & \text{for a lower one-sided test} \\ \frac{n_1}{4} \left( 1 + \sqrt{1 + \frac{4c}{An_1}} \right)^2 & \text{for an upper one-sided test} \end{cases}$$
(4)

For a two-sided hypothesis, n is computed by iteratively solving the two-sided power equation given in (1) using the one-sided estimates as starting values. Without continuity correction, (3) is used to obtain the one-sided sample estimates for the starting values; with a continuity correction, (4) is used.

### References

Agresti, A. 2013. Categorical Data Analysis. 3rd ed. Hoboken, NJ: Wiley.

Armitage, P. 1955. Tests for linear trends in proportions and frequencies. Biometrics 11: 375-386.

- Cochran, W. G. 1954. Some methods for strengthening the common chi-squared tests. Biometrics 10: 417-451.
- Graubard, B. I., and E. L. Korn. 1987. Choice of column scores for testing independence in ordered 2XK contingency tables. *Biometrics* 43: 471–476.
- Lachin, J. M. 2011. Biostatistical Methods: The Assessment of Relative Risks. 2nd ed. Hoboken, NJ: Wiley.
- Nam, J. 1987. A simple approximation for calculating sample sizes for detecting linear trend in proportions. *Biometrics* 43: 701–705.
- Sribney, W. M. 1996. FAQ: Does Stata provide a test for trend? http://www.stata.com/support/faqs/statistics/test-for-trend.

### Also see

- [PSS] power Power and sample-size analysis for hypothesis tests
- [PSS] **power, graph** Graph results from the power command
- [PSS] **power, table** Produce table of results from the power command

[PSS] Glossary

[R] epitab — Tables for epidemiologists

[R] logit — Logistic regression, reporting coefficients